
A project that integrates Phoenix with Ecto, implementing all relevant protocols.
Installation
You can install phoenix_ecto by adding it to your list of dependencies in mix.exs:
def deps do
  [{:phoenix_ecto, "~> 4.0"}]
end
The Phoenix <-> Ecto integration
Thanks to Elixir protocols, the integration between Phoenix and Ecto is simply a matter of implementing a handful of protocols. We provide the following implementations:
	Phoenix.HTML.FormData protocol for Ecto.Changeset
	Phoenix.HTML.Safe protocol for Decimal
	Plug.Exception protocol for the relevant Ecto exceptions

Concurrent browser tests
This library also provides a plug called Phoenix.Ecto.SQL.Sandbox that allows developers to run acceptance tests powered by headless browsers such as ChromeDriver and Selenium concurrently. If you are not familiar with Ecto's SQL sandbox, we recommend you to first get acquainted with it by reading Ecto.Adapters.SQL.Sandbox documentation.
To enable concurrent acceptance tests, make sure you are using PostgreSQL and follow the instructions below:
	Set a flag to enable the sandbox in config/test.exs:
 config :your_app, sql_sandbox: true

	And use the flag to conditionally add the plug to lib/your_app/endpoint.ex:
 if Application.get_env(:your_app, :sql_sandbox) do
   plug Phoenix.Ecto.SQL.Sandbox
 end
 Make sure that this is placed before the line plug YourApp.Router (or any other plug that may access the database).


You can now checkout a sandboxed connection and pass the connection information to an acceptance testing tool like Hound or Wallaby.
Hound
To write concurrent acceptance tests with Hound, first add it as a dependency to your mix.exs:
{:hound, "~> 1.0"}
Make sure to start it at the top of your test/test_helper.exs:
{:ok, _} = Application.ensure_all_started(:hound)
Then add the following to your test case (or case template):
use Hound.Helpers

setup tags do
  pid = Ecto.Adapters.SQL.Sandbox.start_owner!(YourApp.Repo, shared: not tags[:async])
  on_exit(fn -> Ecto.Adapters.SQL.Sandbox.stop_owner(pid) end)
  metadata = Phoenix.Ecto.SQL.Sandbox.metadata_for(YourApp.Repo, pid)
  Hound.start_session(metadata: metadata)
  :ok
end
Hound supports multiple drivers like Chrome, Firefox, etc but it does not support concurrent tests under PhantomJS (the default).
Wallaby
To write concurrent acceptance tests with Wallaby, first add it as a dependency to your mix.exs:
{:wallaby, "~> 0.25", only: :test}
Wallaby can take care of setting up the Ecto Sandbox for you if you use use Wallaby.Feature in your test module.
defmodule MyAppWeb.PageFeature do
  use ExUnit.Case, async: true
  use Wallaby.Feature

  feature "shows some text", %{session: session} do
    session
    |> visit("/home")
    |> assert_text("Hello world!")
  end
end
If you don't use Wallaby.Feature, you can add the following to your test case (or case template):
use Wallaby.DSL

setup tags do
  pid = Ecto.Adapters.SQL.Sandbox.start_owner!(YourApp.Repo, shared: not tags[:async])
  on_exit(fn -> Ecto.Adapters.SQL.Sandbox.stop_owner(pid) end)
  metadata = Phoenix.Ecto.SQL.Sandbox.metadata_for(YourApp.Repo, pid)
  {:ok, session} = Wallaby.start_session(metadata: metadata)
end
Wallaby currently supports ChromeDriver and Selenium, allowing testing in almost any browser.
Configuration
The Plug.Exception implementations for Ecto exceptions may be disabled by including the error in the mix configuration.
config :phoenix_ecto,
  exclude_ecto_exceptions_from_plug: [Ecto.NoResultsError]
Copyright and License
Copyright (c) 2015, Chris McCord.
Phoenix/Ecto source code is licensed under the MIT License.

Integrates Phoenix with Ecto.
It implements many protocols that makes it easier to use
Ecto with Phoenix either when working with HTML or JSON.

A plug that does some checks on your application repos.
Checks if the storage is up (database is created) or if there are any pending migrations.
Both checks can raise an error if the conditions are not met.
Plug options
	:otp_app - name of the application which the repos are fetched from
	:migration_paths - a function that accepts a repo and returns a migration directory, or a list of migration directories, that is used to check for pending migrations
	:migration_lock - the locking strategy used by the Ecto Adapter when checking for pending migrations. Set to false to disable migration locks.

A plug to allow concurrent, transactional acceptance tests with
Ecto.Adapters.SQL.Sandbox.
Example
This plug should only be used during tests. First, set a flag to
enable it in config/test.exs:
config :your_app, sql_sandbox: true
And use the flag to conditionally add the plug to lib/your_app/endpoint.ex:
if Application.compile_env(:your_app, :sql_sandbox) do
  plug Phoenix.Ecto.SQL.Sandbox
end
It's important that this is at the top of endpoint.ex, before any other plugs.
Then, within an acceptance test, checkout a sandboxed connection as before.
Use metadata_for/2 helper to get the session metadata to that will allow access
to the test's connection. In general lines, you would write this:
setup tags do
  pid = Ecto.Adapters.SQL.Sandbox.start_owner!(YourApp.Repo, shared: not tags[:async])
  on_exit(fn -> Ecto.Adapters.SQL.Sandbox.stop_owner(pid) end)
  metadata_header = Phoenix.Ecto.SQL.Sandbox.metadata_for(YourApp.Repo, pid)
  # pass the metadata to the acceptance test library
  :ok
end
You can follow the instructions for Wallaby here.
Acceptance tests with channels
To support channels, in addition the above, you need to make it so each channel
is allowed within the sandbox. The first step is to access the relevant header
metadata.
To do so, you must declare that you want to pass connection information to your
socket. This is typically the user agent header, but it can be a custom x-header
too:
socket "/path", Socket,
  websocket: [connect_info: [:user_agent, …]]

socket "/path", Socket,
  websocket: [connect_info: [:x_headers, …]]
Now use the c:Phoenix.Socket.connect/3 callback to access the header and
store it in the socket:
# user_socket.ex
def connect(_params, socket, connect_info) do
  {:ok, assign(socket, :phoenix_ecto_sandbox, connect_info.user_agent)}
end
Or if you are using a custom header:
Enum.find_value(connect_info.x_headers, fn
  {"x-my-custom-header", val} -> val
  _ -> nil
end)
This stores the value on the socket, so it can be available to all of your
channels for allowing the sandbox.
# room_channel.ex
def join("room:lobby", _payload, socket) do
  allow_ecto_sandbox(socket)
  {:ok, socket}
end

# This is a great function to extract to a helper module
defp allow_ecto_sandbox(socket) do
  Phoenix.Ecto.SQL.Sandbox.allow(
    socket.assigns.phoenix_ecto_sandbox,
    Ecto.Adapters.SQL.Sandbox
  )
end
allow/2 needs to be manually called once for each channel, at best directly
at the start of c:Phoenix.Channel.join/3.
Acceptance tests with LiveViews
LiveViews can be supported in a similar fashion to channels. First declare the
:user_agent (or a custom header) in your live socket configuration in endpoint.ex:
socket "/live", Phoenix.LiveView.Socket,
  websocket: [connect_info: [:user_agent, session: @session_options]]
Now you can use the on_mount/4 callback to check the header and assign the sandbox:
defmodule MyApp.LiveAcceptance do
  def on_mount(:default, _params, _session, socket) do
    %{assigns: %{phoenix_ecto_sandbox: metadata}} =
      assign_new(socket, :phoenix_ecto_sandbox, fn ->
        if connected?(socket), do: get_connect_info(socket, :user_agent)
      end)

    Phoenix.Ecto.SQL.Sandbox.allow(metadata, Ecto.Adapters.SQL.Sandbox)
  end
end
Now, in your web.ex file, you can invoke this callback for all of your
LiveViews if the sandbox configuration, defined at the beginning of the
documentation, is enabled:
def live_view do
  quote do
    use Phoenix.LiveView
    # ...

    if Application.compile_env(:your_app, :sql_sandbox) do
      on_mount MyApp.LiveAcceptance
    end

    # ...
  end
end
Concurrent end-to-end tests with external clients
Concurrent and transactional tests for external HTTP clients is supported,
allowing for complete end-to-end tests. This is useful for cases such as
JavaScript test suites for single page applications that exercise the
Phoenix endpoint for end-to-end test setup and teardown. To enable this,
you can expose a sandbox route on the Phoenix.Ecto.SQL.Sandbox plug by
providing the :at, and :repo options. For example:
plug Phoenix.Ecto.SQL.Sandbox,
  at: "/sandbox",
  repo: MyApp.Repo,
  timeout: 15_000 # the default
This would expose a route at "/sandbox" for the given repo where
external clients send POST requests to spawn a new sandbox session,
and DELETE requests to stop an active sandbox session. By default,
the external client is expected to pass up the "user-agent" header
containing serialized sandbox metadata returned from the POST request,
but this value may customized with the :header option.
Finally, make sure your repository mode is set either to :manual
or {:shared, self()} before the external client starts. This is
typically done by default in your test/test_helper.exs, but you
may need to do it explicitly depending on your setup:
Ecto.Adapters.SQL.Sandbox.mode(MyApp.Repo, :manual)

        Decodes encoded metadata back into map generated from metadata_for/2.

          @spec metadata_for(Ecto.Repo.t() | [Ecto.Repo.t()], pid(), keyword()) :: map()


      


Returns metadata to establish a sandbox for.
The metadata is then passed via user-agent/headers to browsers.
Upon request, the Phoenix.Ecto.SQL.Sandbox plug will decode
the header and allow the request process under the sandbox.

  
    
    Options
  


	:trap_exit - if the browser being used for integration
testing navigates away from a page or aborts a AJAX request
while the request process is talking to the database, it
will corrupt the database connection and make the test fail.
Therefore, to avoid intermittent tests, we recommend trapping
exits in the request process, so all database connections shut
down cleanly. You can disable this behaviour by setting the
option to false.

Spawns a sandbox session to checkout a connection for a remote client.

iex> {:ok, _owner_pid, metadata} = start_child(MyApp.Repo)

  



  
    
      
      Link to this function
    
    stop(owner)


      
       
       View Source
     


  


  

Stops a sandbox session holding a connection for a remote client.

  
    
    Examples
  


iex> {:ok, owner_pid, metadata} = start_child(MyApp.Repo)
iex> :ok = stop(owner_pid)

